
Abstract: Systematic extortion involves a long term parasitic relationship between the criminal and the victim.
Game theory analysis has provided insight into the choices of individual hypothetical criminal and victim pairs.
In thispaperwepresent anagent-basedmodel soas toextend theanalysis to the relationshipbetweenextorters
and other potential victims. The model is developed in two stages, the first to be closest to game theory, the
second one making the decision informed by the social environment of the victim. The agent-based model
shows the importance of social aspects for the functioning of extortion rackets.

Introduction

1.1 Extortion is the demand for money using the threat of violence. An extortion racket is the continuous, regular
and systematic extortion of several victims by a criminal or (more usually) a criminal organisation. Extortion is
di�erent from other property crime such as burglary or fraud because of the long term, explicit and exploita-
tive relationship between perpetrator and victim. It is also di�erent from blackmail because extortion is o�en
endemic in a community or even society.

1.2 There are somewell-known examples of protection or extortion rackets. Probably the best known is the Italian
Mafia, both in Italy (in particular the South and Sicily) and in migration countries such as the US and Germany
(Gambetta 1994). In 2012 the estimated turnover of the Italian Mafia was thought to be around £116bn, which,
at 7% of GDP, makes it Italy’s biggest business1. By definition, extortion racketeering is an essential feature
of a Mafia Type Organisation (MTO), distinguishing MTOs from other organised crime groups (La Spina 2008).
Other MTOs include the Russian Mafia (Smith and Varese 2001), motorcycle gangs such as the Hell’s Angels
and Bandidos, the Yakuza and a variety of gangs and drug cartels in Central and South America (Anzola and
Elsenbroich (2015)).

1.3 One of the most prevalent scientific approaches to extortion rackets is a formalisation of the interdependent
choices of extorters and entrepreneurs in game theory (such as Gambetta 1994; Konrad and Skaperdas 1998;
Smith and Varese 2001). This formalisation adopts the rational choice paradigm underlying much of criminol-
ogy (Becker 1968). But whereas usually the focus is on the decision of the criminal to commit a crime given a
probability of punishment if caught, here the focus is the victim’s decision to pay given the probability of being
punished by an extorter. This punishment probability is influenced by exogenous factors such as the presence
of police (so that Mafiosi are deterred from punishing) and, in some cases, the presence of fakers. Fakers are
agents posing as Mafiosi and demanding payment but whowill not punish resistors, due to fear of detection by
police and real Mafiosi. We discuss these approaches in more detail in Section 2.

1.4 Inspired by this approachwe develop an agent-basedmodel (ABM) of extortion racketeering to understand the
dynamic interdependencies between punishment (for both deterrence and reputation) and levels of acquies-
cence in a society. The agent behaviour in ourmodel also uses the rational paradigm. However, we take advan-
tage of the dynamic nature of ABMs to allow thememory of a punishment to fade over time. This contrasts with
the Game Theory approach of examining the rational decision for di�erent fixed probability values.

1.5 The ABM is presented in two steps. The first is a discussion of an implementation that, like Game Theory, fo-
cusses only on informationobtained throughdirect experience; perceivedprobability of punishment is high im-
mediately a�er being punished but reduces over time. Themodel is then extended to allow the entrepreneur’s
decision to comply or resist an extorter to also depend on events in their neighbourhood. This model is an in-
tuitive implementation of reputation for the Mafia. The neighbourhood model requires much lower levels of
punishment, which is more consistent with real life extortion racket phenomena.

1.6 This paper is structured as follows: In Section 2 we discussmajor game theoretic approaches to extortion rack-
ets (developed in Konrad and Skaperdas 1998; Gambetta 1994; Smith and Varese 2001). In Section 3we discuss
existing agent-based models of extortion and related dynamics, such as reputation and deterrence. In Section
4 an agent-based model inspired by the game theoretic framework from Section 2 is presented and analysed.
In Section 5 this agent-based model is extended to incorporate social aspects of Mafia reputation and the two
approaches compared. In Section 6 we conclude and discuss future work.

Game Theoretic Approaches to Extortion Rackets

2.1 The extortion situation intuitively reads as an iterated decision tree. The extorter demands money from an en-
trepreneur, the entrepreneur decides to pay or not to pay, and then the extorter decides whether to punish.
The demand and decision occurs multiple times between a paired extorter and entrepreneur. In the game the-
oretic literature on extortion rackets (Konrad and Skaperdas 1998; Gambetta 1988, 1994; Varese 1994, 2001), the
payment decision depends on the entrepreneur’s expectation of being punished for refusal to pay. If the payment
demand is smaller than the expected loss fromdamage (that is, the probability of punishmentmultiplied by the
damage cost), the entrepreneur will pay the extorter. Usually protection demands are small and punishments
potentially very high (such as arson attacks on the shop taking away the livelihood). This means the impor-
tant variable to assess for the entrepreneur is the probability of being punished. Two broad approaches can be
distinguished in the literature.

2.2 The first approach (Konrad and Skaperdas 1998) treats the probability of punishment as endogenous, arising
from the e�orts invested by extorters and police. The game described in (Konrad and Skaperdas 1998) for-
malises the interdependent choice between the three actors: the extorter, the entrepreneur and the police.
The approach sets out to find equilibria for investment into punishment by extorters and e�orts by the police
to undermine extortion. Entrepreneurs simply calculate the payo� of compliance or resistance, given levels of
police presence and e�ort compared to extorter aggression.

2.3 On the other hand, (Gambetta 1994; Smith and Varese 2001) have levels of police presence as an exogenous
variable, focussing instead on the extorter-victim interaction. The game originates with (Gambetta 1988) who
uses it as a setting to discuss reputation, signalling and piracy of symbols. Extorters are split into two types, the
real Mafiosi and fakers. Mafiosi establish a reputation for punishing resistance and fakers can free-ride on this
reputation if they manage to signal as belonging to the real Mafia. Entrepreneurs decide whether to pay or not
depending on their individual expected punishment probability calculated fromwhether they think an extorter
is a faker or not, given knowledge of the level of police presence. High police presence lowers the probability of
a Mafioso to punish and also the probability of fakers to extort.

2.4 The game has been formally analysed in detail in (Smith and Varese 2001). Police presence is either high or
low, determined probabilistically. From the construction of the various risks and costs, both Mafiosi and fakers
demand in some situations, but only Mafiosi punish resistance and only when police presence is low.

2.5 The single iterationMafia-faker game results in three equilibria. Region 1 describes situationswith a highproba-
bility of high police presence. Entrepreneurs never pay as they feel safe frompunishment due to the high police
presence. Thus, fakers do not even attempt extortion as there is no benefit. Only Mafiosi demand, and they
only punish refusal when police presence is low (a low probability occurrence). Region 2 has a lower proba-
bility of high police presence and hence higher probability that refusal will result in punishment if the extorter
is a Mafioso. In this region, the cost of punishment to the entrepreneur is high and entrepreneurs believe that
extorters are (mostly) Mafiosi, so they pay the relatively small demands of all extorters. This provides an op-
portunity for fakers to free-ride on the Mafia reputation and the incentives for fakers partly depend on the risk
of encroaching on Mafia territory. As entrepreneurs always pay, they are unable to use their own experience to
recognise whether extorters are real Mafiosi.

2.6 Region 3 is characterised by uncertainty and leads to the most interesting behaviour. While Mafiosi always de-
mand, fakers alsodemandsometimes. Entrepreneurspay sometimes, dependingon the interplayof their belief
that an extorter is a faker and the levels of police presence. Refusal only leads to punishmentwhere the extorter

is a Mafioso and police presence is low. Fakers are more likely to make demands when their expectation of be-
ing paid is higher; because entrepreneurs believe that extorters are Mafiosi, police presence is low (so there is
risk of punishment) or demand amount is small (so lower cost than the risk of punishment).

2.7 These regionsarealso important indescribing the repeatedversionof thegame, analysed indetail in (Smithand
Varese 2001) and the inspiration for the ABM presented here. While the three regions are substantially similar
to the single iteration situation, behaviour near the borders is changed once future interactions are also con-
sidered. In particular, there is a stronger incentive for Mafiosi and fakers to extort and punish, and a stronger
incentive for entrepreneurs to reject the demands. However, entrepreneurs also recognise the stronger incen-
tive for extorters to punish, which creates a countering influence to pay. For extorters, the expensive decision to
risk detection (by police or Mafia for fakers, or by police for Mafia) is potentially repaid with two payments. For
entrepreneurs, the risk of punishment for refusal is being judged against the cost of two payments. That is, an
expensive risk is spread because it establishes behaviour for two turns, or acts as a "signal of type" (Smith and
Varese 2001, pg 367). Instead of the three regions, these competing incentives lead to six equilibria in the two
iteration game. In contrast with the single turn game, even fakers will punish a refusal in one of the regions.

2.8 As well as identifying these equilibria, (Smith and Varese 2001) noted counterintuitive results concerning lev-
els of violence. In particular, violence is a sign of incomplete control by the Mafia; violence only occurs where
entrepreneurs refuse a demand. This may arise either because they know police presence is generally high (re-
gion 1) or believe the extorter could be a faker (region 3). With low police presence, there is no violence once
the Mafia has established control (by eliminating fakers). Violence then erupts from shocks to the system, such
as changes in police presence or the size of demands.

Agent-Based Modelling Approaches to Extortion Racketeering

3.1 In this paper we develop an ABM of extortion racketeering using the Mafiosi-faker game described above as a
conceptual base.

3.2 There are three published ABMs directly concerning extortion racket dynamics. Sonzogni et al. (2011) presents a
model to investigate the social and economic influences on extortion of entrepreneurs, recruitment to a crimi-
nal organisation and judicial punishment on the social entrenchment of extortion rackets. It includes individual
labourmarket conditions and risk perception, combinedwith a social influencemechanismbased on imitating
economic success. The model is used to investigate the combined e�ect of state punishment of, and recruit-
ment to, a criminal organisation.

3.3 Nardin (2016) describes a model focussed on extortion racketeering in Sicily and southern Italy. Based on the
cognitive agent architecture Emil-A, developed to understand normative behaviour, the model represents in
particular cultural dimensions of extortion racketeering. It explicitly investigates a change in culture in Italy of
both law enforcement and bottom-up behaviour change through normative communication.

3.4 Troitzsch (2014, 2015) presents amodel to investigate the e�ects of extortion racketeering on the distribution of
wealth in society and the interactions between state, civil society and criminal organisations regarding compli-
ance, resistance, denunciation and prosecution of criminals. Law enforcement plays amajor role in this model
as policemen are in charge of arresting Mafiosi and interact with entrepreneurs that denounce extortion. The
model simulates e�ects of extortion racket at the systems level, using thresholds and probabilities. This imple-
mentation allows a systematic exploration of the parameters potentially relevant to an extortion racket, such
as propensity to pay. Themodel suggests that themain influence on the success of an extortion racket is the ef-
fectiveness of prosecutionof extorters by the state, followedby theprobability of victims todenounceextorters.

3.5 Agent-basedmodellinghasalsobeenused toexaminedeterrence in social compliancephenomenamoregener-
ally, outside of the extortion racket application. Mafia punishment to induce payment through fear is a specific
application of deterrence models, which have been widely used in ABM, for example to engender pro-social
behaviour with utility maximising agents in models of social norms (Elsenbroich and Gilbert 2013, Chapter 3).
ABM has implemented deterrence with punishment in a variety of ways, with di�erent punishment structures
(e.g. central or distributed) and various cognitive mechanisms (e.g. memory of past experiences or perceiving
whether others are punished).

3.6 A model of tax evasion developed and analysed in (van Baal 2004) includes punishment in a way that is most
similar to its role in extortion dynamics as we discuss here. The model uses the cellular automata approach,
with each location on a grid representing one agent. Agents situated next to each other are neighbours and

in each others’ social networks. At each step, agents decide whether to evade tax or not. Tax evasion without
getting caught has a positive payo�, while norm compliance has a payo� of zero and being found out to evade
tax is sanctionedbyanegativepayo� (a cost). A lawenforcement agency checksagents at random, representing
the probability of getting caught through audit. The agent bases its decision onwhether the expected payo� of
crime is greater than the expected payo� for compliance. The expected payo� is calculated, following Becker
(1968), by weighing the potential payo� of transgression against the probability and cost of getting caught.

3.7 Agents commit a crime if they perceive crime as profitable. The payo� and cost of transgression is known to
the agent. The probability of being audited is estimated by the agent from its social network. Themodel shows
that higher sanctions and higher audit probabilities lead to higher levels of compliance.

3.8 This paper draws on the game theoretic and ABMdeterrence literatures to operationalise aMafia-faker concep-
tual framework for ABM. The narrative of the Italian Mafia is transferred from the original game theory article
(Smith and Varese (2001)) and what our model sets out to show is that amodel of an extortion racket that does
not take into account the neighbourhood e�ects on deterrence (and Mafia reputation by proxy) misses one of
the most important aspects of extortion racketeering. The model allows for a detailed analysis of the role of
punishment to be extended to a community setting, where entrepreneurs learn from the experience of others,
as well as their own.

An Agent-Based Model of the Mafia-Faker-Game

4.1 We present an ABM of the interaction game between Mafiosi, fakers and entrepreneurs. The model consists of
twokindsof agents: extorters andentrepreneurs. Extorters canbe realMafiosi or fakers. Thedi�erencebetween
Mafiosi and fakers is that Mafiosi punish resistance whereas fakers, whilst making extortion demands just like
Mafiosi, do not punish. Police presence is deliberately le� out of this model as it is an exogenous influence on
punishment probability (following Smith and Varese 2001) rather than an active agent. This means the model
focusses on the equilibrium regions in which entrepreneursmake decisions uncertain of the situation (Regions
2 and 3 in Section 2).

4.2 Entrepreneurs are the targets of extortion demands by both fakers and Mafiosi. Extorted entrepreneurs de-
cide in each simulation step whether they pay the pizzo (the amount of the payment requested) or start re-
sisting. An ABM to model the dynamics of extortion rackets needs to implement the conditions under which
entrepreneurs pay or resist (decisionmechanism), punishment of an entrepreneur by theMafia in case of resis-
tance and how this punishment influences an entrepreneur’s future decisionmaking. Consequently, factors to
be operationalised include an extorter’s reputation to punish and the entrepreneur’s fear.

4.3 If all extorters are real Mafiosi (and police are ine�ective), anyone approached should pay as punishment costs
exceeds the pizzo and the probability of punishment is very high. In the game theoretic analysis the faker intro-
duces incomplete knowledge into the game (Gambetta 1994), replacing the certainty of being punished with a
probability. This probability is the essential feature of theMafia-faker game. The decisions of the entrepreneurs
depend on how they assess the probability of being punished and the expected damages arising from punish-
ment by the Mafia.

Decision:

{
If pP ×D ≥ E, pay
otherwise, refuse,

(1)

where pP is the probability of punishment,D the size of the possible damage caused by punishment andE the
size of the pizzo.

4.4 Game theory itself does not provide a cognitive mechanism through which an agent might obtain the punish-
ment probability. Instead, expected behaviour is presented for di�erent regions of the probability space. If the
agent assesses the probability at some value, it adopts a specific behaviour (pay or resist) based on that assess-
ment. The adopted behaviour is the option calculated to maximise the expected payo�, the probability of an
outcomemultiplied by the payo� of that outcome for each behaviour choice.

4.5 In an ABM, the agent perceives its environment and chooses a behaviour based on a combination of that en-
vironment and its own characteristics. In the model presented here, the perceived punishment probability is
determined by an entrepreneur’s past experience on refusing to pay. If the refused extorter is a Mafioso, the
entrepreneur may get punished, if it is a faker, there will be no punishment. If an entrepreneur is punished, its
perceivedpunishment probability becomes 1. This is thendividedby 1.1 (representing adecay of 0.09) each step
the entrepreneur is not punished as the agent ‘forgets’ about the trauma of being punished and increasingly

Parameter Range

Probability Update Type Memory
Mafiosi only 3, 5, 7, 10
Mixed Mafiosi:fakers 7:3, 5:5, 3:7
Extorter Radius 1, 2, 3, 4, 5, 10
Punishment Cost 1, 300, 1000

Table 1: Experimental Design: Memory Model.

wishes to reassess the situation. Input parameters in this model are the territory of an extorter (the extorter-
radius), some financial inputs (the amount of money requested (pizzo) and the damaged incurred by an en-
trepreneur if it is punished (damages)), the overall number of entrepreneurs and the number of Mafiosi and
fakers.

4.6 One additional input needs to be discussed in a little more detail. The ability of Mafiosi to punish resistant en-
trepreneurs is dependent on thewealth of theMafia being in excess of the cost of punishing. In the experiments
the cost is varied at 1, 300 and 1000. Although a punishment has some actual monetary cost associated, like
the gasoline needed for the arson attack, the cost represents the disutility associated with the risk of being ar-
rested by the police. Thus a level of 1 means the Mafia has almost total impunity, 300means there is some cost
associated and 1000 stands for a relatively high level of authority interference. These particular values were
chosen to demonstrate di�erent patterns in the model results. At 300, punishment is regular but not constant,
and Mafiosi were successfully undermined at 1000 (corresponding to high police presence).

4.7 Themodel is initialisedwith 1000 entrepreneurs randomly locatedona 33×33 grid. The appropriate number of
extorters are located in a fixed circle of radius 5 around the centre of the grid. The extorter-radiusdetermines the
area of each extorter’s territory. Territories overlap, but each entrepreneur is extorted by at most one extorter.
Extorters are initialised with 1000 units of capital. Entrepreneurs have a monthly income, normally distributed
with a mean of 500 and a standard deviation of 100. Every 30 time steps entrepreneurs are paid their monthly
income and extorters are paid their pizzo by the entrepreneurs they extort.

4.8 Every time step, each extorter chooses a random entrepreneur within their territory and if the entrepreneur is
not already extorted, theymake an extortion request (if the entrepreneur is already extorted nothing happens).
In response, the entrepreneur decides whether to pay. In addition, entrepreneurs who are already being ex-
torted reassess whether they will continue to pay. In both cases, the decision to pay is based on the expected
probability of punishment as described at Equation 1 above. If the extorter is a Mafioso and has su�icient re-
sources, it chooses an entrepreneur to punish from those resisting.

4.9 Experimentation focusses on the analysis of how the extorter radius and ratio of Mafiosi or fakers together in-
fluence levels of resistance and the number of punishments. The financial inputs are kept constant, the pizzo at
100 and the potential damages an entrepreneur expects at 1,000,000. We vary the number of extorters overall,
the ratio ofMafiosi and fakers, the extorter radius and, as discussed above, the cost of punishment for theMafia.
Table 1 describes the parameter combinations, with 126 simulations.

Results and Analysis

5.1 Analysis focussed on how resistance and punishment are influenced by changes in the extorter population,
extorter radius, and cost of punishment. To eliminate run-in e�ects, model results ignore the first 500 time
steps and are reported as the average value over the following 1000 time steps.

5.2 The model produces a broad range of levels of resistance, from 1% to 100% of entrepreneurs approached by
extorters. Resistance is clearly related to the extorter radius, cost of punishment for the Mafia and the presence
of fakers (see Figures 1 and 2). A full interpretation also relies on the interplay between levels of resistance and
punishment.

5.3 Focussingon theMafiaonly scenarios (Figure 1), as thecostof punishment increasesand their territory expands,
the Mafia have too few resources to retain control and resistance increases. With a punishment cost of 1000,
each punishment consumes the resources obtained from 10 acquiescent entrepreneurs and there is insu�icient
revenue to regularly punish resisters. Unpunished resistance leads to lower revenue and worsens the problem
(from the Mafia perspective), with the consequent overall behaviour of no punishment and total resistance for

Figure 1: Proportion of entrepreneurs resisting extortion attempts (le�) and the proportion of Mafia who punish
(right). Eachpanel presents the results for oneof the fourMafia only scenarios, with 3, 5, 7 or 10Mafiosi extorting
(MafiaM3F0,MafiaM5F0,MafiaM7F0 andMafiaM10F0 respectively) andpresents the average results over 1,000
time steps for simulations that vary by cost of punishment and the radius of extorter territories.

any extorter radius. At the other extreme, only a single acquiescent entrepreneur is required to allow aMafioso
to punish at every time step if the punishment cost is 1. The Mafia are able to punish as frequently as required
and maintain control until their territory becomes too large. At extorter radius of 10, the constraint of a single
punishment per time step (per Mafioso) leads to imperfect control. Despite continual punishment (Figure 1,
right), a significant proportion of the entrepreneurs are resisting at any one time step (Figure 1, le�) in those
scenarios with 3, 5 or 7 Mafiosi. At 10 Mafiosi, however, the territories overlap su�iciently that there are fewer
entrepreneurs per Mafioso and they are able to maintain control and reduce the punishment level.

5.4 The presence of fakers increases entrepreneur resistance and, for the scenarios with punishment cost of 300,
increases the punishments that must be made by the Mafia (Figure 2). The increased resistance is straightfor-
ward, many entrepreneurs are in the territory of fakers but not Mafia and, once they realise they are not being
punished, they continue to resist.

5.5 The punishment e�ect is more interesting. The introduction of fakers makes no di�erence in the number of
punishments per Mafioso per time step where the punishment cost is 1 or 1,000. At a cost of 1, Mafia are able to
punish whenever they like and adding fakers does not change the frequency they need to punish to maintain
control. At a cost of 1,000, Mafia do not have the resources to punish once their initial allocation has been
consumed and adding fakers does not add to their resources so they are also unable to punish in the faker
scenarios.

5.6 The additional punishments in the faker scenarios for punishment cost of 300 arise from the particular inter-
action of entrepreneur density and di�erent financial elements in the experiments. The combination of decay
rate (1− 1

1.1 ) in assessment of punishment probability, pizzo size (100) and damages arising from punishment
(1,000,000)means thatanentrepreneur resists 102 timestepsa�erbeingpunished (byequation 1), duringwhich
time theymake3 (or sometimes4) payments and therefore generally provide su�icient resources topay for only
one punishment. Having been punished, the extorter will acquiesce the next time they are approached andwill
continue to pay for the remainder of 102 time steps a�er the punishment. With di�erent parameter variables,
the same behaviour would occur but at a di�erent punishment cost. The general principle is that at a moder-
ate punishment cost, the territory of each Mafioso must be small enough to support regular revisits. For the
Mafia only scenarios, the territory is too large once extorter radius reaches 4, and financial resources constrain
the number of punishments (compare cost 300 cells to cost 1 cells at Figure 1 right). However, when sharing en-
trepreneurswith fakers, theMafia territories are reducedand theyhave su�icient resources topunishwhenever

Figure 2: Proportion of entrepreneurs resisting extortion attempts (le�) and the proportion ofMafiawhopunish
(right). Each panel presents the results for one of the four scenarios with 10 extorters, with 3, 5, 7 or 10 Mafiosi
extorting and the remainder fakers (Mafia M3F7, Mafia M5F5, Mafia M7F3 and Mafia M10F0 respectively) and
presents theaverage results over 1,000 timesteps for simulations that varyby costof punishmentand the radius
of extorter territories.

they require for radius 4 and 5 but not radius 10 (Figure 2 right).
5.7 The model we have presented recreates some of the results of the game theoretic model. In particular, re-

sistance is high when there is a high punishment cost (equivalent to high police presence) and, at lower pun-
ishment cost, resistance increases as the proportion of fakers increases. In addition, violence (punishment)
increases as the Mafia try to exert control over a greater number of entrepreneurs. However this model falls
short of replicating one of themost stable stylised facts of extortion rackets: high levels of compliance with low
levels of punishment. This combination only occurs for small extorter territories. In this model, deterrence of
entrepreneurs is implemented in a purely individualistic way using their memory of being punished, and the
Mafia must immediately punish resisters to keep control of a territory. This is costly and leads to Mafiosi ‘run-
ning out of resources’ and loss of control. In the next section, we extend the model to implement a reputation
mechanism for Mafiosi.

An Agent-Based Model of Mafia Reputation

6.1 The presented ABM applies a strongly individualistic interpretation of deterrence, chosen to stay as close as
possible to the game theoreticmodel of Smith and Varese (2001). In this sectionwe develop an ABM that allows
entrepreneurs to also gain information from the experience of their neighbours. That is, an entrepreneur is not
only deterred by what happened to him or herself but also by experiences of others nearby. The individualistic
memory based calculation of the punishment probability does not take this social aspect of deterrence into
account.

6.2 In the extended model, the memory based calculation is replaced by a mechanism where the punishment
probability is determined by the observation of punishment of others as well as personal experience. If an en-
trepreneur observes a punishment in a certain neighbourhood, its perceived punishment probability becomes
1, decaying each step the entrepreneur does not observe a punishment. The input parameters in thismodel are
the same as above, i.e. extorter-radius, pizzo, damages and cost of punishment, number of Mafiosi, number of
fakers, number of entrepreneurs, but in addition entrepreneurs have a neighbourhood-radius, determining the
size of the neighbourhood in which observed punishments a�ect their perceived punishment probability.

Figure 3: The two figures represent a situation with one Mafioso, two fakers (purple and grey large dots) and
two entrepreneurs (small dots). The memory-dependent punishment probability is shown in (a) and the ex-
tended model with neighbourhood e�ects in (b). The circles around them visualise the extorter radius and (in
(b)) neighbourhood radius. Red entrepreneurs pay the pizzo, green entrepreneurs resist paying. For the mem-
orymodel (a), the red entrepreneurmight bepaying the faker because it is in the intersectionof the realMafioso
and the faker, and has been punished previously by the Mafioso, but the green entrepreneur will always resist
as it will not personally experience punishment. For the neighbourhood model (b), the Mafioso punishment
is within the neighbourhood radius (large circle) of the isolated faker and can therefore provide a deterrence
e�ect on the entrepreneur that would have resisted in the memory model. This provides an extended reach of
the deterrent e�ect from the punishments of the real Mafioso and hence free-riding by appropriately located
fakers.

Parameter Range
Probability Update Type Neighbourhood
Mafiosi only 3, 5, 7, 10
Mixed Mafiosi:fakers 7:3, 5:5, 3:7
Extorter Radius 1, 2, 3, 4, 5, 10
Neighbourhood Radius 1, 2, 3, 4, 5, 10
Punishment Cost 1, 1000

Table 2: Experimental Design — Neighbourhood Model

6.3 The neighbourhood based calculation is expected to enhance the deterrence of each punishment as it not only
deters the punished entrepreneur but also those in that are closer than the neighbourhood radius (see Figure
3). This socially enhanced deterrence of punishments should lead to:

1. lower levels of resistance, in particular as the neighbourhood radius increases;

2. lower levels of punishment for similar levels of resistance; and

3. easier free-riding for fakers on the Mafia’s reputation.

6.4 We run similar scenarios as in the memory case discussed in Section 4 but, in addition to varying the extorter-
radius, we also vary the neighbourhood-radius. Punishment cost is set to 1,000 for all experiments, which is the
value atwhich thepreviousmodel hadnopunishments and fully resistant entrepreneurs because theMafia had
insu�icient resources tomaintain control. Table 2describes theparameter combinations,with 504 simulations.

Results and Analysis

7.1 As for Section 4, the presented results are the average value over 1,000 time steps, a�er discarding the first 500
to eliminate run in e�ects. In figures 1 and 2, the 1,000 punishment cost scenarios had full resistance (yellow)

Figure 4: Proportion of entrepreneurs resisting extortion attempts (le�) and the proportion ofMafiawhopunish
(right). Eachpanel presents the results for oneof the fourMafia only scenarios, with 3, 5, 7 or 10Mafiosi extorting
(MafiaM3F0,MafiaM5F0,MafiaM7F0 andMafiaM10F0 respectively) andpresents the average results over 1,000
time steps for simulations that vary by neighbourhood radius (the distance over which an entrepreneur can
observe punishments) and the radius of extorter territories. The punishment cost is fixed at 1,000, which is the
same value for the top row of each panel in Figure 1.

and no punishments (blue) for all extorter combination and extorter radius combinations (top lines of each
panel). In the extended model (figures 4 and 6), this same pattern is observed for neighbourhood radius up to
2, but begins to break down as neighbourhood radius increases.

7.2 For theMafia only scenarios (Figure 4), the constraining factor is the resources available for punishment, as can
be seen by the existence of at least some resistance combined with punishment levels below 1. This constraint
controls the model behaviour; a larger extorter territory allows greater revenue but also requires more pun-
ishments. As the neighbourhood radius increases, each punishment potentially influences the behaviour of a
larger number of entrepreneurs and therefore allows control over a greater territory.

7.3 For the particular experiment values, resistance generally decreases as the number of Mafiosi increases. How-
ever, this is not true for extorter radius of 2, where resistance is high for the 3 and 10 Mafiosi scenarios and
lower for the 5 and 7 scenarios. At 3 and 10 Mafiosi, there are too few entrepreneurs in each territory to provide
resources for punishment because the territories are, respectively, too small or too overlapping.

7.4 For all Mafia only scenarios, resistance decreases and punishment increases as extorter radius increases, and
resistance decreases with little change in punishment as neighbourhood radius increases. As extorter radius
increases for a fixed neighbourhood size, the pizzo revenue increases and enablesmore punishment. Provided
theentrepreneurs are able toobserve su�icient nearbypunishments (neighbourhood radiusof at least 3), these
punishments induce enoughacquiescence to fund further punishments. As neighbourhood radius increases for
a fixed extorter territory, each punishment is observed bymore entrepreneurs and resistance decreases, which
increases extorter revenue and funds a small increase in punishments.

7.5 A di�erent pattern emerges where the Mafia don’t need to rely on multiple entrepreneurs to provide the re-
sources for punishment. With punishment cost set to 1, the number of punishments consistently decreases
as the neighbourhood increases (Figure 5, right). That is, the need to punish consistently decreases if there is
no resource constraint. Resistance shows a di�erent pattern, decreasing for initial increases in neighbourhood
size, but then increasing where the neighbourhood radius is larger than the extorter radius (Figure 5, le�). This
increase arises because all the entrepreneurs are influenced by the same punishment and thus become resis-
tant again in the same time step. Where there are few (and therefore distant) extorters, the synchrony leads to

JASSS, 19(4) 8, 2016 http://jasss.soc.surrey.ac.uk/19/4/8.html Doi: 10.18564/jasss.3223



Figure 5: Proportion of entrepreneurs resisting extortion attempts (le�) and the proportion ofMafiawhopunish
(right). The punishment cost is fixed at 1, which allowsMafiosi to punish every time step if appropriate.Both the
original model (neighbourhood radius = 0) and the extendedmodel (neighbourhood radius > 1) are presented.
Each panel presents the results for one of the fourMafia only scenarios, with 3, 5, 7 or 10Mafiosi extorting (Mafia
M3F0, Mafia M5F0, Mafia M7F0 and Mafia M10F0 respectively) and presents the average results over 1,000 time
steps for simulations that vary by neighbourhood radius and the radius of extorter territories.

high resistance at regular time steps, which is resolved with a single punishment on each occasion.

7.6 As with the initial model, the presence of fakers increases entrepreneur resistance but there is no e�ect on
punishment levels compared to the Mafia only scenarios (Figure 6). Some entrepreneurs are in the territory of
fakers but not Mafia and, if their neighbourhood radius does not extend to the Mafia territory, they cannot ob-
serve punishments and continue to resist. As the neighbourhood radius increases, there is a smaller di�erence
between the Mafia only andmixed Mafia / faker scenarios.

7.7 One consequence of the Mafia punishing in the mixed Mafia and faker scenarios is that entrepreneurs may ac-
quiesce to a faker’s extortion attempt. As fakers do not bear the cost of punishment, fakers who successfully
use the Mafia’s reputation to extort can accumulate greater wealth than Mafia. At time step 1,000, the average
wealth of fakers is between 1.2 and 101.7 times that of theMafia. This ratio increases as extorter radius increases
but has no clear relationship with neighbourhood radius (Figure 7).

Conclusion and Future Work

8.1 We have presented two ABMs of extortion rackets inspired by the game theoretic analysis of extortion by Gam-
betta (1994) and Smith and Varese (2001). This approach uses a three player game consisting of a Mafioso, a
faker and an entrepreneur, with the latter having some belief about probability of being punished for resisting
extorter attempts. The analysis identified regions of punishment probability space with di�erent entrepreneur
decisions about resisting and extorter decisions to punish. A key result is that there are regions with high com-
pliance and low punishment, with high levels of violence indicating loss of Mafia control. However, the game
theory framework does not provide a mechanism by which entrepreneurs can assess punishment probability,
it is exogenous.

8.2 The ABM were implemented with the same three active kinds of agents; Mafiosi, fakers and entrepreneurs.
The entrepreneur decision-mechanism calculates the utility of complying or resisting, determined by the en-
trepreneur’s belief about the probability of being punished and the impact of the punishment.

Figure 6: Proportion of entrepreneurs resisting extortion attempts (le�) and the proportion of Mafia who pun-
ish (right). Each panel presents the results for one of the four scenarios with 10 extorters, with 3, 5, 7 or 10
Mafiosi extorting and the remainder fakers (Mafia M3F7, Mafia M5F5 and Mafia M7F3 and Mafia M10F0 respec-
tively) andpresents the average results over 1,000 time steps for simulations that vary byneighbourhood radius
(the distance over which an entrepreneur can observe punishments) and the radius of extorter territories. The
punishment cost is fixed at 1,000, which is the same value for the top row of each panel in Figure 2.

Figure 7: The average faker wealth divided by the average Mafioso wealth at time step 1,000. Both the original
model (neighbourhood radius = 0) and the extended model (neighbourhood radius > 1) are presented. Each
panel presents the results for one of the three scenarios with 10 extorters, with 3, 5 or 7 Mafiosi extorting and
the remainder fakers (Mafia M3F7, Mafia M5F5 and Mafia M7F3 respectively).

8.3 In the first model the punishment probability belief is calculated using a memory based mechanism; if an ex-
torter is punished, it will have a high punishment probability, slowly decreasing over time. Themodel recreated
most of the game theoretic results, showing how increasing punishment is a sign of Mafia weakness and also
showed the detrimental e�ect fakers have on the Mafia reputation. It failed, however, to recreate one of the
most stable stylised fact of extortion rackets: the combination of low levels of resistance and low levels of pun-
ishment. In order to achieve low resistance levels Mafiosi had to punish constantly. Further, if punishment is

costly (representing risk of police detection), the Mafia may collect insu�icient resources from the acquiescent
entrepreneurs to punish the resisters and a spiral of increasing resistance occurs.

8.4 We then extended the memory based ABM by an alternative decision-mechanism based on observing punish-
ments in the neighbourhood rather than only relying on personal experience. This model implements some
social aspects of deterrence and reputation through social amplification Kasperson et al. (1988) of the e�ect of
each punishment. This model resulted in much more realistic extortion behaviour, where low levels of resis-
tance are paired with low levels of punishment.

8.5 Although thememorymechanisms leads to results with high levels of acquiescence, it is clear that social trans-
mission of reputation renders punishment much more e�ective. One consequence of this extension, however,
is that fakers who are su�iciently close to a Mafioso are able to take advantage of the Mafia propensity to pun-
ish and extort without the cost of punishing. Even with non-overlapping extorter radii, if the neighbourhood
observation radius is large enough, fakers can survive in the shadow of the Mafia.

8.6 Themodels presented are quite simple and abstract, but allow the exploration of some fundamental dynamics
of punishment, reputation and resistance. They suggest that some form of reputation is required in order to
achieve compliancewithout substantial punishment. In futureworkwewill implement extensions on the same
level of abstraction that examine di�erent social structures in which the agents operate. In particular we will
investigate ways in which extortion rackets can be undermined or fought by both bottom up strategies, such as
normative reasoning and collective reasoning spreading resistance and top down interventions such as police
pressure and surveillance.

Appendix: Model Code

breed [entrepreneurs entrepreneur]
breed [extorters extorter]

directed-link-breed [payments payment]
directed-link-breed [demands demand]
directed-link-breed [rejections rejection]

globals [punishments
         cumulative-punishments 
         sum-compliance
         sum-collective
         sum-entrepreneur-wealth
         sum-extorter-wealth
         ]
        
entrepreneurs-own [
      
       monthly-income
       wealth
       demanded?
       extorted?
       my-extorter
       acquiescent
       punished?
       ever-punished?
       punishment-probability
       damage-suffered
       my-neighbours
       times-punished
]

extorters-own [
       mafia?
       wealth
       extortees
       demanding?
       extortion-radius
       refused 
       punishment-probability
       my-payments
       my-income
       my-extorter-radius
       my-rejections
]

    

to setup 
  clear-all
;  random-seed 0
  
  if number-of-fakers > number-of-extorters
  [ print "Too many fakers, exceeds total extorters" stop ]
  ask patches [set pcolor white]
  
  create-entrepreneurs number-of-entrepreneurs
  create-extorters number-of-extorters
  
  ask entrepreneurs [ setxy random-xcor random-ycor
                      if any? entrepreneurs-here [fd 1]
                     
                      set color blue
                      set monthly-income random-normal 500 100
                      set wealth monthly-income
                      set extorted? false
                      set demanded? false
                      set punished? false
                      set ever-punished? false
                      set acquiescent 0
                      set punishment-probability random-float initial-punishment-probability
                      set my-neighbours entrepreneurs in-radius neighbourhood-radius 
                      set shape "circle 2"
                    ]
         
  layout-circle extorters 15
  ask extorters     [ set color 117
                      set shape "circle" 
                      set extortion-radius extorter-radius
                      set label-color black 
                      set mafia? TRUE
                      set wealth random-normal 1500 400
                      set my-payments count my-in-payments
                      set my-rejections count my-in-rejections
                     ]

Simplified Model - page 1

  ask n-of number-of-fakers extorters [ set mafia? FALSE 
                                        set color grey ]
  
  ask payments      [ set color red ]
  ask demands       [ set color yellow ]
  
  reset-ticks
end

to go 
  establish-punishment-probability
  extort
  if ticks > 0 and ticks mod 31 = 0 [payday]
  resist
  clear-punishments
  punish

  update-measurements
  tick
  if ticks > 3000 [ stop ] 
end

;;;; now to the procedures

;;;external source of money coming into the system and distribution to extorters
to payday
  ask entrepreneurs
  [ set wealth wealth + monthly-income
    if extorted? [ set wealth wealth - (count my-out-payments * pizzo ) ]
  ]
  
  ask extorters
  [ set wealth wealth + (count my-in-payments * pizzo)
    set my-income count my-in-payments * pizzo
  ]
end

to establish-punishment-probability
  if punish-prob-update = "Memory"
  [ ask entrepreneurs 
    [ ifelse punished?
      [ set punishment-probability 1 ]
      [ set punishment-probability punishment-probability / 1.1]
    ]
  ]

  if punish-prob-update = "Neighbourhood"
  [ ask entrepreneurs
    [ let joinset (turtle-set self my-neighbours)
      let punished-neighbours joinset with [ punished? ]
      ifelse count punished-neighbours >= 1
      [ set punishment-probability 1 ]
      [ set punishment-probability punishment-probability / 1.1]
    ]
  ]
end

to extort
  ;; each extorter selects a random entrepreneur, if not currently extorted then demand
  ask extorters
  [ set extortion-radius extorter-radius
    let potential-extortees entrepreneurs in-radius extortion-radius with [ not extorted? ]
    if any? potential-extortees
    [ ask one-of potential-extortees
      [ set my-extorter myself
        create-demand-from myself
        set color yellow
        ask my-in-demands [ set color yellow ]
        make-decision
      ]
    set size count my-in-payments / 10
    set my-extorter-radius count my-in-payments 
    ]
  ]
end  

to make-decision
 let price-to-pay damages
 ifelse (punishment-probability  *  price-to-pay) >= pizzo [pay] [refuse]
end

to pay 
    set acquiescent 1
    set extorted? true

Simplified Model - page 2

    set punished? false
    ask my-extorter [ create-payment-from myself ] 
    ask payments [set color red]
    ask my-in-demands [die]
    ask my-out-rejections [die]
    set color red
end
  
to refuse
    set acquiescent -1
    set extorted? false
    set demanded? false
    ask my-in-demands [die]
    ask my-extorter
    [ set refused refused + 1
      create-rejection-from myself
      ask my-in-rejections [ set color green ]
    ]
    set color green
end

to resist
 ;; checks with those acquiescent to determine if want to change to resist 
 ask entrepreneurs
 [ let price-to-pay damages
    if extorted? and (punishment-probability * price-to-pay) <=  pizzo
    [ refuse
      ask my-out-payments [die]
    ]
 ]
end

to punish 
  let real-mafiosi extorters with [mafia?]
  ask real-mafiosi
  [ let my-wealth [wealth] of self
    let current-cost cost
    let price-to-pay random damages
    if current-cost < my-wealth
    [ if count in-rejection-neighbors >= 1
      [ ask one-of in-rejection-neighbors
        [ set wealth wealth - price-to-pay
          set color black
          set punished? true
          set ever-punished? true
          set times-punished times-punished + 1
          set damage-suffered price-to-pay
        ]
        set wealth wealth - current-cost
        set color 113
        set punishments punishments + 1
        set cumulative-punishments cumulative-punishments + 1
      ]
    ]
  ]
end                        
                       
to  clear-punishments
   ask entrepreneurs with [ punished? ]
   [ set punished? false
     set color yellow ]
   ask extorters with [mafia?] [set color 117]
   set punishments  0
end

to  update-measurements 
  set sum-compliance sum-compliance + count entrepreneurs with [acquiescent = 1]
  set sum-extorter-wealth sum-extorter-wealth + mean [wealth] of extorters
  set sum-entrepreneur-wealth sum-entrepreneur-wealth + mean [wealth] of entrepreneurs
  ask extorters [set my-payments count my-in-payments]
end

              
 
 

Simplified Model - page 3

